Using the Timer/Event Counter in the HT49 MCU Series

D/N：HA0027E

Introduction

This application note gives an introduction how to use the timer/event counter in the HT49 series of MCUs.

The clock source for the HT49 series can be summarized in the following table:

	HT49R30A-1	HT49R50A-1/HT49R70A-1	
	Timer/Event Counter	**Timer/Event Counter 0**	**Timer/Event Counter 1**
Options	1. System clock 2. System clock/4	1. System clock 2. System clock/4	1. System clock 2. Timer/event counter 0 overflow signal 3. Time base overflow signal
Program control bit option	External clock	1. External clock 2. RTC overflow signal	External clock

For all timers, if the external clock input is chosen, the user can implement external counting, time interval measurement or pulse width measurement functions. The generation of a precise time base can also be implemented.

Using registers TMR0/TMR1 and TMR0C/TMR1C, the program can be used to control the mode and initial value of the Timer/event counters. The definition of the register structure and each bit are:

Bit No.	Name	Functional Description
0~2	—	No definition, read “0”
3	TE	Define the active edge of the timer/event counter TMR0 (0=rising edge, 1=falling edge)
4	TON	Enable/disable counter (0=disable, 1=enable)
5	TS	Internal clock source option (0=RTC output, 1=system clock or instruction clock)
6 7	TM0 TM1	Define operation -TM1, TM0 01=counter mode - external clock) 10=counter mode - system clock 11=pulse width detect mode - external clock 00=not defined

TMR0C Register

Bit No.	Name	Functional Description
0~2	—	Without definition, read as “0”
3	TE	Define the function edge of the timer/event counter TMR1 (0=rising edge, 1=falling edge)
4	TON	Enable / disable timer/event counter (0=disable, 1=enable)
5	TS	Enternal clock source option (0=clock decided by option, 1=instruction clock)
6 7	TM0 TM1	Define the operation method 01=counter mode - external clock 10=counter mode - internal clock 11=pulse width detect mode - external clock 00=not defined

TMR1C register

Some program examples follow showing how to use the Timer/event counters:

Program Example

→ **Using the Timer Mode**

The following example shows how to use the Timer/event counter in the timer mode taking the HT49R50A-1 as an example and with the system frequency/4 as the timer clock source. Set TM0 and TM1 as 10 for the timer mode, and TE as 0 for rising edge trigger.

- Option
  - RC type oscillator with a frequency 400kHz
  - Counter 0 options choose system frequency/4 as clock source
  - PA port as CMOS

The counter initial value is 9CH, which will produce a 500Hz square wave on the PA port.

The formula for the initial value of the counter:

Set the initial value as X

From the certain condition T=1 / f = 1 / (4 × 105 / 4) = 1×10-5

(28－X) × 1 × 10-5= 10-3, which gives a value of X = 156

In 8-bit binary this is: 10011100, therefore the TMR0 initial value should be 9C

- Program Description

  Each time the Timer/event counter overflows, the interrupt operation will cause the PA output lines to invert. In this way a 500Hz clock can be created which can be seen with an oscilloscope on PA.

- Program List

```
;------ ------------------------------------------------------------
;FILE NAME：timeruse1.asm
;Writer：Chin Juan
;Purpose：an introduction to using the Timer/event Counter in the HT49 series
;;;
;-------------------------------------------------------------------
include      ht49r50a-1.inc
code .section at 0h 'code'
     org     00h
             jmp  start
     org     0ch
             cpl  pa               ;interrupt handling, invert PA output
                                   ;to generate frequency
             reti
     org     20h
     start:
```

```
                clr  pa              ;clear PA
                set  intc0.0         ;enable master interrupt
                set  intc0.3         ;enable interrupt timer/event counter 0
                                     ;interrupt
                mov  a,9Ch           ;set the initial value of the counter
                mov  tmr0,a
                mov  a,0ah           ;setup the timer mode with a clock source
                                     ;of external clock/4
                mov  tmr0c,a
                set  tmr0c.4         ;counter enabled
                jmp  $
end
```

→ **Using the Counter Mode**

The following example shows how the Timer/event counter is used in the counter mode. Set the TM0, TM1 bits to the value of 01 to setup the Timer/event counter in the external counter mode. With the TE bit set to 0, the Timer/event counter will be active on the rising edge of the external signal.

- Program List

```
;--------------------------------------------------------------
;FILE NAME：countermode.asm
;Writer：Chin Juan
;Purpose：an introduction to the usage of the HT49 series timer couner mode
;;;
;------------- ------------------------------------------------
include      ht49r50a-1.inc
data .section 'data'
     count db ?
code .section 'code'
     org     00h
             jmp  start
             org  0ch
             jmp  count1          ;jump to interrupt service subroutine
                                  ;count1
     org  20h
     start:
             clr  count           ;clear the count register
             mov  a,09h           ;set the Timer/event counter interrupt enable
             mov  intc0,a
             mov  a,40h           ;setup in counter mode, rising edge trigger
             mov  tmr0c,a
             mov  a,0cdh          ;set the initial value of the counter
             mov  tmr0,a
             set  tmr0c.4         ;enable counter
             jmp  $
     count1:                      ;interrupt service subroutine
             inc  count           ;interrupt occurs, count add 1
```

```
                mov  a,count          ;move count to the accumulator
                sub  a,64h            ;check if the interrupt time has exceeded 100
                sz   acc
                reti
                clr  count            ;if 100 times, clear count
                reti
end
```

→ **Using the Pulse Width Measurement Mode**

The following example shows how the Timer/event counter is used in the pulse width measurement mode. Set the TM0, TM1 bits to 11, to setup the Timer/event counter in the pulse width measurement mode. With the TE bit set to 1, the high level width will be measured.

When the Timer/event counter is in the pulse width measurement mode, with the TON and TE bits equal to 1, if the TMR0/TMR1 external pins receive a rising edge edge signal (note that if TE = 0, it will be a falling edge signal), the counter will begin counting until the TMR0/TMR1 pin returns to its original level at which point the TON bit will be cleared to ″0″. Note that it is only in this mode that the TON bit will be automatically cleared to ″0″. For the other two modes, the TON bit can be cleared using instructions. When the counter stops counting, the measured result will still be kept in the Timer/event counter. Therefore, the pulse width measurement mode can only detect one pulse. If the TON bit is reset, when the TMR0/TMR1 pin receives another pulse, the detect cycle will continue. When in this mode, the Timer/event counter will not count according to the logical levels but rather by the signal edges. Once the counter overflows, it will be automatically reloaded from the value already stored in the Timer/event counter, and will also issue a timer interrupt request in the same way as the timer mode.

- Options
  Choose a crystal oscillator frequency of 4000kHz
- Test result
  The pulse measurement range of the program is 8Hz~500kHz
- Program List

```
;----------------------------------------------------------------
;FILE NAME：pulse.asm
;Writer：Chin Juan
;Purpose：Using the HT49 series Timer/event counter in the pulse width measurement
mode
;;;
;----------------------------------------------------------------
include ht49r50a-1.inc
data .section 'data'
```

```
        count db ?
code .section 'code'
        org     00h
                jmp  start
        org  0ch
                inc  count              ;If the pulse width exceeds 0FFH, the counter
                                        ;overflows, add 1 to the count register
                reti
        org     20h
        start:
                clr  count              ;initialise the value
                clr  intc0
                mov  a,09h
                mov  intc0,a            ;enable master and Timer/event counter 0 interrupt
                mov  a,00h
                mov  tmr0,a             ;set the initial value of the counter
                mov  a,0ffh
                      mov  tmr0c,a      ; pulse width measurement mode - detect high level
        loop:
                snz  tmr0c.4            ;check counter enable bit for reset
                jmp  count1             ;over (counter = 0) jump to count1
                jmp  loop               ;not over (counter = 1), stay in the loop
        count1:
                mov  a,tmr0
                jmp  $
end
```